Problem:
Score : 400 points

Problem Statement

There is a directed graph with N vertices and M edges.
The i-th edge (1≤i≤M) points from vertex a_i to vertex b_i, and has a weight c_i.
We will play the following single-player game using this graph and a piece.

Initially, the piece is placed at vertex 1, and the score of the player is set to 0.
The player can move the piece as follows:

When the piece is placed at vertex a_i, move the piece along the i-th edge to vertex b_i. After this move, the score of the player is increased by c_i.

The player can end the game only when the piece is placed at vertex N.
The given graph guarantees that it is possible to traverse from vertex 1 to vertex N.

When the player acts optimally to maximize the score at the end of the game, what will the score be?
If it is possible to increase the score indefinitely, print inf.

Constraints

2≤N≤1000

1≤M≤min(N(N-1),2000)

1≤a_i,b_i≤N (1≤i≤M)

a_i≠b_i (1≤i≤M)

a_i≠a_j or b_i≠b_j (1≤i<j≤M)

-10^9≤c_i≤10^9 (1≤i≤M)

c_i is an integer.

In the given graph, there exists a path from vertex 1 to vertex N.

Input

Input is given from Standard Input in the following format:

N M
a_1 b_1 c_1
a_2 b_2 c_2
:
a_M b_M c_M

Output

Print the maximum possible score at the end of the game, if it is finite. If it is possible to increase the score indefinitely, print inf.

Sample Input 1

3 3
1 2 4
2 3 3
1 3 5

Sample Output 1

7

There are two ways to move the piece to vertex N=3:

vertex 1 → vertex 2 → vertex 3 : score 4+3=7

vertex 1 → vertex 3 : score 5

Thus, the maximum possible score at the end of the game is 7.

Sample Input 2

2 2
1 2 1
2 1 1

Sample Output 2

inf

It is possible to increase the score indefinitely by alternating between vertex 1 and 2.

Sample Input 3

6 5
1 2 -1000000000
2 3 -1000000000
3 4 -1000000000
4 5 -1000000000
5 6 -1000000000

Sample Output 3

-5000000000

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

import Control.Monad
import Data.IntMap.Strict ((!)) 
import qualified Data.IntMap.Strict as M 
import qualified Data.Set as S 
import qualified Data.ByteString.Char8 as B 
import qualified Data.Array.IArray as A 
import Data.Maybe 

type Vertex = Int 
type Weight = Int 
type Edge = (Vertex, (Vertex, Weight)) 
type Graph = A.Array Vertex (S.Set (Vertex, Weight))
type Memo = M.IntMap Weight

buildG :: Int -> [Edge] -> Graph 
buildG n = A.accumArray (flip S.insert) S.empty (1, n) 

from :: Graph -> Vertex -> S.Set (Vertex, Weight) 
from g = (g A.!) 

readDirectedEdge :: B.ByteString -> [Edge] 
readDirectedEdge = map ((\[s,t,w] -> (s,(t,w))) . map readVertex . B.words) . B.lines 

readVertex :: B.ByteString -> Vertex 
readVertex = fst . fromJust . B.readInt

main :: IO ()
main = do
  [n,_] <- map read . words <$> getLine
  es <- readDirectedEdge <$> B.getContents
  let g = buildG n es
  putStrLn $ maybe "inf" show (score n g)

score :: Int -> Graph -> Maybe Weight
score n g = guard (im1!n == im2!n) >> return (im1!n)
  where
    im0 = M.singleton 1 0
    im1 = bellman_ford g n im0
    im2 = bellman_ford g (n + 1) im1

bellman_ford :: Graph -> Int -> Memo -> Memo
bellman_ford g n im
  | n == 1 = im
  | otherwise = bellman_ford g (n - 1) (M.foldrWithKey (move g) im im)

move :: Graph -> Vertex -> Weight -> Memo -> Memo
move g s aw im = S.foldr update im (from g s)
  where 
    update (t,w) acc
      | M.notMember t acc || (acc!t) < aw + w = M.insert t (aw + w) acc
      | otherwise = acc